
Fundamentos de la programación y la informática
Examen de ejercicios de laboratorios. 13 de enero de 2025

URJC. Grados en ingeniería aeroespacial. Turno de tarde

Ejercicio único (10 puntos)
Escribe un programa en Pascal llamado diagonal.pas que cumpla la siguiente especificación:

1. Estará escrito a partir de un esqueleto que encontrarás en el aula virtual, en evaluación, es-
queleto_examen. Completa este esqueleto añadiendo todo lo que necesites, pero no deberías
modificar nada de lo que aparece en el esqueleto.

2. El programa manejará dos matrices cuadradas, llamadas matriz_1 y matriz_2. Sus elemen-
tos serán números reales.

3. matriz_1 se generará de forma aleatoria, lanzando un dado de 20 caras. Naturalmente un
dado devuelve valores enteros, así que matriz_1 tendrá números reales, pero con parte
decimal siempre nula.

4. El programa calculará la media de la diagonal principal de matriz_1.

5. matriz_2 será una matriz con los siguientes valores:

En su diagonal principal, todos los valores serán iguales: la media aritmética de la
diagonal principal de matriz_1.
En el resto de valores, matriz_2 será igual a matriz_1.

6. El programa mostrará en pantalla matriz_1 y la media de la diagonal principal de matriz_1.

7. El programa mostrará en pantalla matriz_2.

8. El programa escribirá matriz_1 en el fichero matriz_1.txt.

9. El programa escribirá matriz_2 en el fichero matriz_2.txt.

Ejemplo de salida en pantalla:

Matriz_1:
12.00 16.00 15.00 2.00
16.00 6.00 16.00 12.00
16.00 1.00 3.00 9.00
11.00 20.00 20.00 7.00

Media diagonal principal:7.00

Matriz_2:
7.00 16.00 15.00 2.00

16.00 7.00 16.00 12.00
16.00 1.00 7.00 9.00
11.00 20.00 20.00 7.00

Observaciones
Se denomina diagonal principal de una matriz cuadrada al conjunto de elementos cuyo
índice de la fila es igual al índice de la columna. Por ejemplo, en la siguiente matriz los
elementos de la diagonal principal contienen el carácter X, y todos los demás, el carácter
punto.

X . . .
. X . .
. . X .
. . . X

Obviamente, este programa tiene que seguir funcionando si modificamos el valor de las
constantes Filas y Columnas, no es válido que escribas un programa que solo funciona si el
número de filas es exactamente 4.

Recuerda que si el programa no compila, su nota será nula.

Esqueleto proporcionado
{$mode objfpc}{$H-}{$R+}{$T+}{$Q+}{$V+}{$D+}{$X-}{$warnings on}

program diagonal;
const

Filas = 4;
Columnas = 4;

type
TipoMatriz = array[1..Filas, 1..Columnas] of real;

function tira_dado(caras_dado:integer):integer;
begin

result := random(caras_dado) + 1;
end;

procedure inicia_matriz(var matriz:TipoMatriz);
var

i,j : integer;
const

CarasDado = 20;
begin

for i := 1 to Filas do
for j:= 1 to Columnas do

matriz[i,j] := tira_dado(CarasDado);
end;

procedure escribe_matriz(matriz:TipoMatriz);
var i,j: integer;
begin

for i := 1 to Filas do begin
for j:= 1 to Columnas do

write(matriz[i,j]:7:2);
writeln;

end;
end;

Solución
{$mode objfpc}{$H-}{$R+}{$T+}{$Q+}{$V+}{$D+}{$X-}{$warnings on}
program diagonal;
const

Filas = 4;
Columnas = 4;

type
TipoMatriz = array[1..Filas, 1..Columnas] of real;

function tira_dado(caras_dado:integer):integer;
begin

result := random(caras_dado) + 1;
end;

procedure inicia_matriz(var matriz:TipoMatriz);
var

i,j : integer;
const

CarasDado = 20;
begin

for i := 1 to Filas do
for j:= 1 to Columnas do

matriz[i,j] := tira_dado(CarasDado);
end;

procedure escribe_matriz(matriz:TipoMatriz);
var i,j: integer;
begin

for i := 1 to Filas do begin
for j:= 1 to Columnas do

write(matriz[i,j]:7:2);
writeln;

end;
end;

procedure escribe_matriz_fichero(matriz:TipoMatriz; nombre_fichero:string);
var i,j: integer;

fichero: text;
begin

assign(fichero, nombre_fichero); // Ponemos nombre al fichero
rewrite(fichero); // Apertura en modo escritura
for i := 1 to Filas do begin

for j:= 1 to Columnas do
write(fichero,matriz[i,j]:7:2);

writeln(fichero);
end;
close(fichero);

end;

function media_diag_principal(matriz:TipoMatriz):real;
var

i: integer;
sumatorio: real;

begin
sumatorio := 0;
for i:= 1 to Filas do begin

sumatorio := sumatorio + matriz[i,i];
end;
result := sumatorio/Filas;

end;

function reemplaza_diag_principal(matriz:TipoMatriz; nuevo_valor:real):TipoMatriz;
// Recibe una matriz. Devuelve otra igual, excepto en la diagonal
// principal, donde escribe el nuevo valor.
var

i,j: integer;
begin

for i:=1 to Filas do begin
for j:= 1 to Columnas do begin

if i<>j then
result[i,j] := matriz[i,j]

else
result[i,j] := nuevo_valor;

end;
end

end;

var
matriz_1, matriz_2 : TipoMatriz;

begin
randomize();
inicia_matriz(matriz_1);
writeln('Matriz_1:');
escribe_matriz(matriz_1);
writeln();
writeln('Media diagonal principal:', media_diag_principal(matriz_1):4:2);
writeln();

writeln('Matriz_2:');
matriz_2 := reemplaza_diag_principal(matriz_1, media_diag_principal(matriz_1));
escribe_matriz(matriz_2);

escribe_matriz_fichero(matriz_1, 'matriz_1.txt');
escribe_matriz_fichero(matriz_2, 'matriz_2.txt');

end.

